Theory and Computation in the Search for Special Surfaces

Let $\mathbb{K}$ be an algebraically closed field, let $\mathbb{P}^n$ be the n-dimensional projective space over $\mathbb{K}$, and let V and W be two disjoint irreducible projective varieties in $\mathbb{P}^n$. We denote by $J(V, W)$ the union of the lines in $\mathbb{P}^n$ joining V to W. Then $J(V, W)$ is a projective variety. This variety is called the *join* of V and W.

Exercise 1. (a) Write a *Macaulay 2* script for computing the ideal of the join of two given varieties. Let L_1 and L_2 be two skew lines in $\mathbb{P}^4$. Compute the ideal of $J(L_1, L_2)$.
(b) Make a *Macaulay 2* function for computing the ideal of the secant variety to a given variety (use the script you wrote in **Exercise 1**). Use your script to compute the ideal of the secant variety to the Veronese variety in $\mathbb{P}^5$.

Hint. Let $A = [a_0 : \cdots : a_n]$ and $B = [b_0 : \cdots : b_n]$ be points of V and W respectively. Then any point $R = [z_0 : \cdots : z_n]$ of the line passing through A and B is given by

$$\begin{pmatrix} z_0 \\ \vdots \\ z_n \end{pmatrix} = s \begin{pmatrix} a_0 \\ \vdots \\ a_n \end{pmatrix} + t \begin{pmatrix} b_0 \\ \vdots \\ b_n \end{pmatrix} \tag{1}$$

for some $[s : t] \in \mathbb{P}^1$. Let $\{f_1, \ldots, f_l\}$ and $\{g_1, \ldots, g_m\}$ be the generating sets for $I(V)$ and $I(W)$ respectively. Then the ideal defining $J(V, W)$ is obtained by solving the system of equations $f_1 = \cdots = f_l = 0$, $g_1 = \cdots = g_m = 0$ and (1) for $z_0, \ldots, z_{n-1}$ and z_n. By replacing x_i's by a_i' and b_i's, we obtained new ideals $I(V)_a$ in $\mathbb{K}[a_0, \ldots, a_n]$ and $I(W)_b$ in $\mathbb{K}[b_0, \ldots, b_n]$ respectively. Let I be the ideal in the following new ring:

$$\mathbb{K}[a_0, \ldots, a_n, b_0, \ldots, b_n, s, t, z_0, \ldots, z_n]$$

generated by the equations in (1), and let $J = I + I(V)_a + I(W)_b$. Saturate J with respect to the ideal (s, t):

$$J' = J : (s, t)^\infty,$$

because we do not want s and t to vanish at the same time (recall that $[s : t] \in \mathbb{P}^1$). Take the intersection $\bar{J} = J' \cap \mathbb{K}[z_0, \ldots, z_n]$. Then $V(\bar{J}) = J(V, W)$.

Exercise 2. Construct a smooth rational surface X in $\mathbb{P}^4$, which is isomorphic to $\mathbb{P}^2$ blown up in eight points embedded by the linear system of the following type:

$$\left| 4L - 2E_0 - \sum_{i=1}^{7} E_i \right|,$$

where L is the pullback from $\mathbb{P}^2$ a line, while the E_i are the exceptional curves of the blowup. Compute the degree and sectional genus of X.

Exercise 3. Construct the union P of four planes P_0, P_1, P_2 and P_3 in $\mathbb{P}^4$ such that

(a) P_0 intersects each of P_1, P_2 and P_3 in a line;
(b) For $1 \leq i < j \leq 3$, P_i and P_j meet at a single point.

Check that the ideal of U is generated by 7 cubics. Take two general cubics to get the ideal of the surface X linked to U in a $(3,3)$ complete intersection. What is this surface? Any line in P_i, $i \in \{1,2,3\}$, is a 3-secant line to X.

Theorem 1 (Beilinson, 1978)**.** *For any sheaf $\mathcal{F}$ on $\mathbb{P}^n$, there is a complex $\mathcal{K}^{\cdot}$ with*

$$\mathcal{K}^i = \bigoplus H^{i-j}(\mathbb{P}^n, \mathcal{F} \otimes \mathcal{O}_{\mathbb{P}^n}(j)) \otimes \Omega^{-j}(-j)$$

such that

$$H^i(\mathcal{K}^{\cdot}) = \begin{cases} \mathcal{F} & \text{if } i = 0, \\ 0 & \text{otherwise.} \end{cases}$$

Exercise 4. Let $\mathcal{E}$ be a locally free sheaf on $\mathbb{P}^3$ whose cohomology table for $h^i(\mathbb{P}^3, \mathcal{E}(j))$ in the range $-3 \leq j \leq 0$ of twists is of the following type:

i \ j	-3	-2	-1	0
3				
2	2			
1			2	2
0				

$h^i\mathcal{E}(j)$

Use Beilinson's theorem to determine the type of the monad for $\mathcal{E}$. Then construct the differentials of the monad. How do you check that your sheaf is locally free?

Exercise 5. Let X be an elliptic conic bundle in $\mathbb{P}^4$. Compute the image of the adjunction map $\Phi_{|H+K|} : X \to \mathbb{P}^2$. (Use the script for the ideal of X).

You can find more exercises in the last two pages!

```
KK=ZZ/32003
-------------------------------
-- Example: Veronese surface --
-------------------------------

ringP2=KK[x_0..x_2];
ringP4=KK[y_0..y_4];
veronese=basis(2,ringP2)*random(ringP2^6,ringP2^5);
I=ker map(ringP2,ringP4,veronese);
```

```
hilbertPolynomial I

--------------------------------------
-- Example: Veronese surface in P^4 --
--------------------------------------

ringP4=KK[y_0..y_4];
veronese=basis(2,ringP2)*random(ringP2^6,ringP2^5);
I=ker map(ringP2,ringP4,veronese);
hilbertPolynomial I
-- Is V(I) smooth?
singI=minors(codim I,jacobian I)+I;
codim singI
betti I

--------------------------------
-- Example: del Pezzo surface --
--------------------------------

randomPoints=(N)->(
     i:=1;
     idl:=ideal random(ringP2^{0},ringP2^{2:-1});
     while i<N do (
  idl=intersect(idl,ideal random(ringP2^{0},ringP2^{2:-1}));
  i=i+1;
  );
     idl)
points=randomPoints(5);
hilbertPolynomial points
del=gens points*map(source gens points,basis(3,points));
delPezzo=trim ker map(ringP2,ringP4,del);
betti res delPezzo

-------------------------------------------------------------------
-- Example: Veronese surface in P^4 and elliptic quintic scroll --
-------------------------------------------------------------------

veronese=basis(2,ringP2)*random(ringP2^6,ringP2^5);
I=ker map(ringP2,ringP4,veronese);
betti I
V=ideal(gens I*random(source gens I,ringP4^{2:-3}));
J=V:I;
```

```
betti J
X=Proj(ringP4/J);
HH^1(OO_X)

------------------------------------
--Example: nullcorellation bundle --
------------------------------------

sortedBasis=(i,E)->(
 m:=basis(i,E);
 p:=sortColumns (m, MonomialOrder=>Descending);
 m_p);
beilinson1=(e,dege,i,S)->(
  E:=ring e;
  mi:=if i<0 or i>=numgens E then map(E^1,E^0,0)
      else if i === 0 then id_(E^1)
      else sortedBasis(i+1,E);
  r:=i-dege;
  mr:= if r<0 or r>=numgens E then map(E^1,E^0,0)
       else sortedBasis (r+1,E);
  s:=numgens source mr;
  if i===0 and r===0 then substitute(map(E^1,E^1,{{e}}),S)
  else if i>0 and r===i then substitute(e*id_(E^s),S)
  else if i>0 and r===0 then
  (vars S)*substitute(contract(diff(e,mi),transpose mr),S)
  else substitute(contract(diff(e,mi),transpose mr),S));
U=(i,S)->(
     if i<0 or i>=numgens S then S^0
     else if i===0 then S^1
     else coker koszul(i+2,vars S)**S^{i});
beilinson=(o,S)->(
     coldegs:=degrees source o;
     rowdegs:=degrees target o;
     mats:=table(numgens target o, numgens source o,
  (r,c)->(
       rdeg=first rowdegs#r;
       cdeg=first coldegs#c;
       overS=beilinson1 (o_(r,c),cdeg-rdeg,cdeg,S);
       map(U(rdeg,S),U(cdeg,S),overS)));
     if #mats===0 then matrix(S,{{}})
     else matrix(mats));
ringP3=KK[x_0..x_3];
```

```
E=KK[e_0..e_3,SkewCommutative=>true];

-- indecomposable

alpha=map(E^{-1},E^{-3},{{e_0*e_2+e_1*e_3}});
alpha=beilinson(alpha,ringP3);
betti alpha
omega1=U(1,ringP3)
F=prune coker(presentation omega1|map(target presentation omega1,,alpha));
fF=res F;
betti fF
rank F
Ft=syz transpose fF.dd_1;
-- Check F is locally free.
codim minors(2,Ft)
codim minors(3,Ft)

-- decomposable

alpha=map(E^{-1},E^{-3},{{e_0*e_2}});
alpha=beilinson(alpha,ringP3);
betti alpha
omega1=U(1,ringP3)
F=prune coker(presentation omega1|map(target presentation omega1,,alpha));
fF=res F;
betti fF
rank F
Ft=syz transpose fF.dd_1;
codim minors(2,Ft)
codim minors(3,Ft)

------------------------------------
-- Example: elliptic conic bundle --
------------------------------------

ringP4=KK[x_0..x_4];
E=KK[e_0..e_4,SkewCommutative=>true];
beta=random(E^{1:0},E^{1:-1,1:-2})
betti syz beta
alpha=(syz beta)*random(source syz beta,E^{-3});
beta=beilinson(beta,ringP4);
alpha=beilinson(alpha,ringP4);
```

```
F=prune homology(beta,alpha);
fF=res F;
betti fF
phi=presentation F|random(target presentation F,ringP4^{4:-1});
fphit=res (prune coker transpose phi,LengthLimit=>2);
betti fphit
I=ideal fphit.dd_2;
hilbertPolynomial I

------------------------------
-- Example: Bordiga surface --
------------------------------

restart
kk=ZZ/32003
S=kk[a..e]
I=ideal(
     490*a^3+12221*a^2*b+1663*a*b^2-14146*b^3-6390*a^2*c-8829*a*b*c
       +6322*b^2*c-9839*a*c^2-7186*b*c^2-4124*c^3-6141*a^2*d+15804*a*b*d
       -10963*b^2*d-14244*a*c*d+3292*b*c*d+12528*c^2*d+1470*a*d^2
       +9748*b*d^2-4690*c*d^2-11431*d^3-10463*a^2*e+965*a*b*e
       +15018*b^2*e+3510*a*c*e-13207*b*c*e-12332*c^2*e+5496*a*d*e
       -1110*b*d*e-14202*c*d*e+4737*d^2*e+11286*a*e^2+3759*b*e^2
       -12124*c*e^2-8044*d*e^2-8400*e^3,
     -4212*a^3-6354*a^2*b+8359*a*b^2-15620*b^3-7605*a^2*c+14070*a*b*c
       +1533*b^2*c-6538*a*c^2+1929*b*c^2+6483*c^3+2503*a^2*d
       -13769*a*b*d-10716*b^2*d+15442*a*c*d-10241*b*c*d+4046*c^2*d
       -7595*a*d^2-4553*b*d^2-5535*c*d^2-12507*d^3-8300*a^2*e
       +9255*a*b*e-166*b^2*e-3948*a*c*e-887*b*c*e+12166*c^2*e
       -12019*a*d*e-4986*b*d*e-9965*c*d*e-2795*d^2*e-11300*a*e^2
       -9200*b*e^2+2201*c*e^2+771*d*e^2-10878*e^3,
     -13422*a^3-13098*a^2*b+165*a*b^2-15751*b^3-9062*a^2*c
       +3931*a*b*c+10465*b^2*c+3947*a*c^2+15856*b*c^2-13765*c^3
       +12661*a^2*d-759*a*b*d+2367*b^2*d-5578*a*c*d-10985*b*c*d
       +987*c^2*d-9772*a*d^2-1366*b*d^2+2179*c*d^2-13964*d^3
       -11350*a^2*e-1598*a*b*e+2754*b^2*e-9928*a*c*e+13439*b*c*e
       +15037*c^2*e-9065*a*d*e+12521*b*d*e+9348*c*d*e-10519*d^2*e
       -13803*a*e^2+6223*b*e^2-6660*c*e^2+9736*d*e^2+5263*e^3,
     14635*a^3+2059*a^2*b-5482*a*b^2-14747*b^3+1035*a^2*c-4988*a*b*c
       +11895*b^2*c+13195*a*c^2+7097*b*c^2-8995*c^3+1144*a^2*d
       -11329*a*b*d-14481*b^2*d-4991*a*c*d+14404*b*c*d+10242*c^2*d
       -6921*a*d^2-6661*b*d^2-15585*c*d^2+9999*d^3+4902*a^2*e
```

```
        -12029*a*b*e-681*b^2*e-2020*a*c*e+14081*b*c*e+15876*c^2*e
        +7994*a*d*e+7138*b*d*e-3671*c*d*e+11424*d^2*e+380*a*e^2
        -11858*b*e^2+9754*c*e^2+13383*d*e^2+6735*e^3);
betti I
fI=res I;
betti fI
ringP2=kk[s_0..s_2];
ringP4xP2=kk[a..e,s_0..s_2,MonomialOrder=>Eliminate 5];
omegaX=coker transpose fI.dd_2;
phi=sub(vars ringP2, ringP4xP2)*sub(presentation omegaX,ringP4xP2);
adjunction=ideal(oo);
adjunctionsat=saturate(adjunction,ideal(a..e));
J=ideal sub(selectInSubring(1,gens gb oo),ringP2);
codim J
psi=contract(sub(vars S,ringP4xP2),transpose phi)
points=sub(minors(4,psi),ringP2);
hilbertPolynomial points
betti points
```

Lemma 1 (Terracini's lemma)**.** *Let X be a smooth variety in $\mathbb{P}^n$ and let $p_1 \cdots, p_s$ be s generic points on X. Then for a generic $p \in \langle p_1, \cdots, p_s \rangle$,*

$$\dim \sigma_s(X) = \dim \langle T_{p_1}X, \ldots, T_{p_s}X \rangle,$$

where $T_{p_i}X$ is the projectivized tangent space to X in $\mathbb{P}^n$ at p_i.

Let X be a smooth variety in $\mathbb{P}^n$ and let p_1 and p_2 be two generic points on X. Then for a generic $p \in \langle p_1, p_2 \rangle$,

$$\dim \mathrm{Sec}(X) = \dim \langle T_{p_1}X, T_{p_2}X \rangle,$$

where $T_{p_i}X$ is the projectivized tangent space to X in $\mathbb{P}^n$ at p_i.

Exercise 6. Let $X = G(3,6)$ be the Grassmaniann of 3-planes in a 6-dimensional vector space. Use Terracini's lemma to prove that $\mathrm{Sec}(X)$ has the expected dimension.

Hint. Let V be a 6-dimensional vector space with basis $e_0, \ldots, e_5$ and let p_1 and p_2 be the points of X corresponding to $e_0 \wedge e_1 \wedge e_2$ and $e_3 \wedge e_4 \wedge e_5$. Try to prove that

$$\dim \langle T_{p_1}X, T_{p_2}X \rangle = 19.$$

You can use the following fact: Let us denote by W the following subspace of $\bigwedge^3 V$:

$$V \wedge e_1 \wedge e_2 + e_0 \wedge V \wedge e_2 + e_0 \wedge e_1 \wedge V.$$

Note that $T_{p_1}X = \mathbb{P}(W)$. Let $W^\perp = \left\{ w \in \bigwedge^3 V \,|w \wedge e_0 \wedge e_1 \wedge e_2 \right\}$. The orthogonal complement $W^\perp$ can be identified with $[(e_0, e_1, e_2)^2]_3$, where (e_0, e_1, e_2) is the ideal of the exterior algebra $\bigoplus_{i=0}^6 \bigwedge^i V$ generated by e_0, e_1 and e_2 and $[(e_0, e_1, e_2)^2]_3$ is the degree 3 part of $(e_0, e_1, e_2)^2$.

Exercise 7. Let $s \in H^0(\mathbb{P}^3, \mathcal{E}(1))$ be a general section of $\mathcal{E}(1)$. Compute the ideal of the zero locus Z of s. Answer the following questions:

(a) What is the dimension of Z? How about the degree and arithmetic genus of Z?
(b) Is Z smooth?
(c) How many components does Z have?

Exercise 8. Let M be the family of vector bundles whose cohomology tables are as given in **Exercise 4**. Compute the dimension of the Zariski tangent space to M at the point corresponding to the example you constructed in **Exercise 4**.